

**Exercice 1.**

La matrice

$$M = \begin{pmatrix} 6 & -1 & 0 & 1 \\ 4 & 3 & 4 & 1 \\ 0 & 1 & 6 & -1 \\ 4 & -1 & 4 & 5 \end{pmatrix}$$

est-elle diagonalisable ?

**Exercice 2.**

Soit

$$M = \begin{pmatrix} 3 & 1 & 3 & -1 \\ 2 & 2 & 2 & 0 \\ 0 & -1 & 0 & 1 \\ 2 & 0 & 2 & 2 \end{pmatrix}.$$

- Calculer  $M^n$  pour tout  $n \in \mathbb{N}$ .
- Existe-t-il une matrice réelle  $B$  telle que  $M = B^2$  ?

**Exercice 3.**

Soit

$$f(x) = \arctan(x) - \frac{x(1 + ax^2 + bx^4)}{1 + cx^2 + dx^4}, \quad x \in \mathbb{R}.$$

Déterminer les réels  $a, b, c$  et  $d$  de sorte que  $f$  soit un infiniment petit en 0 d'ordre le plus élevé possible, puis majorer  $|f|$  sur l'intervalle  $[-1, 1]$ .

**Exercice 4.**

Soient  $n, m \in \mathbb{N}$ . Calculer le reste de la division euclidienne de  $X^n + 2X^m + 1$  par  $(X-1)(X-2)(X-3)(X-4)$ . Vérifier le résultat pour  $n = 10$  et  $m = 17$  à l'aide de la commande `Remainder`.

**Exercice 5.**

On pose  $P_n = \sum_{k=0}^n \frac{X^k}{k!}$ ,  $n \in \mathbb{N}$ .

- Ecrire une procédure de la variable  $n$  qui représente graphiquement les racines du polynôme  $P_n$  dans le plan complexe. (On pourra utiliser la commande `fsolve`.)
- Conjecturer le nombre de racines réelles de  $P_n$ , puis prouver cette conjecture par récurrence.


**Exercice 6.**

Soit  $A = \begin{pmatrix} 1 & 2 \\ 2 & -1 \end{pmatrix}$ . Dans la classe de conjugaison de  $A$  par les matrices orthogonales trouver toutes les matrices ayant diagonale nulle.

*Remarque :* La classe de conjugaison de  $A$  par les matrices orthogonales est l'ensemble  $\{P^{-1}AP \mid P \in O(2)\}$ .

**Exercice 7.**

Soit  $\mathcal{C}$  un cercle,  $A, B$  deux points distincts sur  $\mathcal{C}$  et  $M$  le milieu de la corde  $[AB]$ . Soient  $[PQ]$  et  $[SR]$  deux autres cordes passant par  $M$ . On note  $C$  (resp.  $D$ ) le point d'intersection de  $[AB]$  avec  $[PS]$  (resp.  $[RQ]$ ).



Démontrer que  $M$  est aussi le milieu de  $[CD]$ . (Faites une preuve par un calcul analytique en coordonnées ; consacrez-vous à l'idée de la preuve et délégez à Maple le travail calculatoire.)

**Exercice 8.**

Soit  $b \geq 2$  un nombre entier. Tout  $n \in \mathbb{N}$  se laisse représenter en base  $b$  :

$$n = \sum_{k=0}^m q_k b^k, \quad q_k \in \llbracket 0, b-1 \rrbracket, \quad q_m \neq 0.$$

Cette écriture est unique et on utilise la notation  $n = [q_m, \dots, q_1, q_0]_b$ . Par exemple  $32 = [1, 0, 1, 2]_3 = [3, 2]_{10}$ .

- Calculs sur papier (sans Maple) :
  - Calculer  $[2, 1, 0, 1, 2]_3$ .
  - Ecrire 100 en base 3, puis en base 4.
- Ecrire une procédure `Maxi` qui prend comme argument un couple d'entiers  $(n, b)$  avec  $n > 0$  et  $b > 1$  et renvoie l'unique  $m \in \mathbb{N}$  vérifiant  $b^m \leq n < b^{m+1}$ .
- Ecrire une procédure `BaseEntier` rendant, en forme de liste, la représentation d'un entier  $n \geq 1$  en base  $b \geq 2$ . (*Indication :* vous pouvez utiliser les commandes `irem` et `iquo` ainsi que la procédure `Maxi`.)

**Exercice 9.**

Identifier la nature des points stationnaires (points où la dérivée s'annule) de la fonction

$$f : \mathbb{R}^2 \rightarrow \mathbb{R}, (x, y) \mapsto 2x^2 - 8xy^2 - \frac{16}{3}y^3 + 8y^2 - 8.$$

Retrouver ces résultats en traçant les lignes de niveau avec l'ordre `contourplot`.

## 1. Solutions

### Solution 1.

Les commandes `Eigenvalues(M)` ou `Eigenvalues(M, output='list')` montrent que 4 est une valeur propre de multiplicité algébrique 2 mais de multiplicité géométrique 1. Donc  $M$  n'est pas diagonalisable.

### Solution 2.

1. La commande `Eigenvalues(M)` montre que 0 et 3 sont des valeurs propres simples et que 3 est une valeur propre de multiplicité algébrique et géométrique 2. Donc  $M$  est diagonalisable sur  $\mathbb{R}$ .

```
> restart: with(LinearAlgebra):
M:=Matrix(4, [3,1,3,-1,2,2,2,0,0,-1,0,1,2,0,2,2]);
Eigenvalues(M);
> P:=Eigenvalues(M)[2];
T:=MatrixInverse(P).M.P;
```

Puisque  $M = PTP^{-1}$  on a  $M^n = PT^nP^{-1}$  et la matrice diagonale  $T^n$  se calcule facilement. Cela permet de calculer les puissances  $M^n$  comme suit.

```
> Tn:=DiagonalMatrix([seq(T[k,k]^(1/2),k=1..4)]);
Mn:=P.Tn.MatrixInverse(P);
```

Attention é ne pas rentrer  $T^n$  é la main ! En fait, Maple sort les valeurs propres dans un ordre au hasard, et si vous fixez un ordre, le programme risque de ne plus marcher lors de la prochaine compilation.

On obtient

$$M^n = \begin{pmatrix} 3^n & 2^{n-1} & 3^n & -2^{n-1} \\ 2(3^n - 2^n) & 2^n & 2(3^n - 2^n) & 0 \\ 0 & -2^{n-1} & 0 & 2^{n-1} \\ 2(3^n - 2^n) & 0 & 2(3^n - 2^n) & 2^n \end{pmatrix}$$

On peut aussi écrire une petite procédure qui donne  $M^n$  en fonction de  $n$ , sans passer par la diagonalisation.

```
> PuissanceM:=proc(n) local A,k :
A:=IdentityMatrix(4):
for k from 1 to n do A:=A.M od: A:
end proc:
> subs(n=10,Mn)-PuissanceM(10); # pour tester
```

2. Oui, une telle matrice « racine carrée » de  $M$  existe. En effet, il est clair qu'il existe une matrice réelle  $C$  telle que  $C^2 = T$ . Alors  $B = PCP^{-1}$  est une racine carrée de  $M$ .

```
> C:=DiagonalMatrix([seq(T[k,k]^(1/2),k=1..4)]);
B:=P.C.MatrixInverse(P);
> g:=x->simplify(x); map(g,B^2-M); #Test
```

On obtient

$$B = \begin{pmatrix} \sqrt{3} & 1/\sqrt{2} & \sqrt{3} & -1/\sqrt{2} \\ 2(\sqrt{3} - \sqrt{2}) & \sqrt{2} & 2(\sqrt{3} - \sqrt{2}) & 0 \\ 0 & -1/\sqrt{2} & 0 & 1/\sqrt{2} \\ 2(\sqrt{3} - \sqrt{2}) & 0 & 2(\sqrt{3} - \sqrt{2}) & \sqrt{2} \end{pmatrix}$$

**REMARQUE** – La nouvelle commande Maple `MatrixPower(M,n)` répond également à toute l'exercice.

### Solution 3.

On utilise la commande `series` pour obtenir un développement limité. Comme on cherche quatre inconnues  $a, b, c, d$  on pousse l'ordre du DL assez loin pour avoir au moins quatre équations en  $a, b, c, d$ . L'ordre 10 suffit.

```
> restart:
f:=x->arctan(x)-x*(1+a*x^2+b*x^4)/(1+c*x^2+d*x^4):
series(f(x),x=0,10);
```

Avec `convert` on convertit la série tronquée en polynôme.

```
> P:=convert(%,polynom):
s:=solve({seq(coeff(P,x,2*k+1),k=1..4)},{a,b,c,d});
```

$$s = \left\{ d = \frac{5}{21}, c = \frac{10}{9}, a = \frac{7}{9}, b = \frac{64}{945} \right\}$$

```
> assign(s); factor(diff(f(x),x));
```

La fonction  $f$  est donc décroissante. Puisque  $f$  est impaire la fonction  $|f|$  atteint son maximum en 1 et  $-1$ .

```
> evalf(f(-1));
```

0.0001874221

Conclusion : La fraction rationnelle  $\frac{x(1 + \frac{7}{9}x^2 + \frac{64}{945}x^4)}{1 + \frac{10}{9}x^2 + \frac{5}{21}x^4}$  est une approximation de la fonction arctangente au voisinage de 0, avec une erreur inférieure à  $1.88 \times 10^{-4}$  sur  $[-1, 1]$ .

**Solution 4.**

On pose  $P := X^n + 2X^m + 1$  et  $S := (X - 1)(X - 2)(X - 3)(X - 4)$ . Il existe un unique couple  $(Q, R) \in \mathbb{R}[X]^2$  tel que

$$P = QS + R, \quad \deg(R) < 4.$$

En évaluant en 1, 2, 3 et 4 on obtient  $R(k) = P(k), k \in [1, 4]$ . Notant  $R = a + bX + cX^2 + dX^3$  on obtient le système linéaire

$$\begin{pmatrix} 1 & 1 & 1 & 1 \\ 1 & 2 & 2^2 & 2^3 \\ 1 & 3 & 3^2 & 3^3 \\ 1 & 4 & 4^2 & 4^3 \end{pmatrix} \begin{pmatrix} a \\ b \\ c \\ d \end{pmatrix} = \begin{pmatrix} P(1) \\ P(2) \\ P(3) \\ P(4) \end{pmatrix}.$$

On remarque qu'il s'agit d'une matrice de Vandermonde inversible ce qui rend plus rapide son écriture dans Maple (évidemment on peut aussi s'en passer et entrer la matrice de manière classique).

```
> restart: with(LinearAlgebra):
P:=X^n+2*X^m+1;
A:=VandermondeMatrix(<1,2,3,4>);
B:=<subs(X=1,P),subs(X=2,P),subs(X=3,P),subs(X=4,P)>;
C:=LinearSolve(A,B);
R:=C[1]+C[2]*X+C[3]*X^2+C[4]*X^3;
```

Vérification pour  $n = 10$  et  $m = 17$ .

```
> with(Algebraic):
subs(n=10,m=17,R)-Remainder(X^10+2*X^17+1,(X-1)*(X-2)*(X-3)*(X-4),X);
```

La différence est bien nulle.

**Solution 5.**

```
1. > restart: with(plots):
P:=proc(n) local S,k;
S:=0:
for k from 0 to n do S:=S+X^k/(k!) od:
end proc:
> Dessin:=proc(n) local L;
L:=[fsolve(P(n)=0,X,complex)];
pointplot([seq([Re(L[k]),Im(L[k])],
k=1..nops(L))],
color=blue, symbol=solidcircle, symbolsize=15):
end proc:
```

> Dessin(2); Dessin(3); Dessin(4); Dessin(5);  
Puisque  $P_n$  est un polynôme réel l'ensemble de ses racines est symétrique par rapport à l'axe des réels. Sur les exemples calculés avec Maple, il paraît que  $P_n$  possède  $n$  racines distinctes; elles sont donc toutes simples.

2. Nous allons prouver que pour tout  $n \in \mathbb{N}$ ,  $P_n$  n'a pas de racine réelle si  $n$  est pair et que  $P_n$  a une unique racine réelle, qui en plus est simple, si  $n$  est impair.  
Initialisation : Le polynôme  $P_0 = 1$  n'a pas de racine

réelle. Le polynôme  $P_1 = 1 + X$  a une unique racine réelle, à savoir  $r_1 = -1$ ; elle est forcément simple.  
Hérédité : Soit  $n \geq 2$ . Considérons d'abord le cas  $n$  pair. Par hypothèse de récurrence,  $P_{n-1}$  admet une unique racine réelle  $a$ . Puisque  $P_{n-1} = P'_n$  et  $\lim_{x \rightarrow \pm\infty} P(x) = +\infty$  cela signifie que  $P_n$  a son minimum global en  $a$ . Alors  $P_n(x) > 0$  pour tout  $x \in \mathbb{R}$  car

$$P_n(a) = P_{n-1}(a) + \frac{a^n}{n!} = \frac{a^n}{n!} > 0.$$

On vient d'utiliser le fait que  $P_n(0) = 1$  pour tout  $n \in \mathbb{N}$  ce qui entraîne que  $a \neq 0$ .

Maintenant considérons le cas où  $n$  est impair. Puisque  $\lim_{x \rightarrow -\infty} P(x) = -\infty$  et  $\lim_{x \rightarrow +\infty} P(x) = +\infty$ , on déduit du théorème des valeurs intermédiaires que  $P_n$  admet une racine réelle. Si  $P_n$  avait deux racines réelles distinctes alors, par le théorème de Rolle, il existait une racine de  $P'_n$ , et comme  $P'_n = P_{n-1}$  ça serait en contradiction avec l'hypothèse de récurrence. Donc  $P_n$  admet une seule racine réelle et elle est simple car  $P'_n$  n'a pas de racine réelle.

**Solution 6.**

Les matrices orthogonales  $2 \times 2$  sont de la forme

$$P = \begin{pmatrix} \cos(a) & -\sin(a) \\ \sin(a) & \cos(a) \end{pmatrix} \quad \text{ou} \quad Q = \begin{pmatrix} \cos(a) & \sin(a) \\ \sin(a) & -\cos(a) \end{pmatrix},$$

avec  $a \in \mathbb{R}$ . On note  $B = {}^tPAP$  et  $C = {}^tQAQ$ . Remarquons la trace d'une matrice est invariante sous conjugaison. Puisque la matrice de départ  $A$  est de trace nulle les matrices  $B$  et  $C$  le sont itou. Donc la diagonale de  $B$

(ou  $C$ ) est nulle si et seulement si son premier coefficient diagonal est nul. Par Maple on trouve que c'est le cas si et seulement si  $B = \pm \begin{pmatrix} 0 & \sqrt{5} \\ \sqrt{5} & 0 \end{pmatrix}$ . Et pareil pour  $C$ .

Pour simplifier je n'ai trouvé rien de mieux que la combinaison `expand(rationalize(radnormal))`... Puisque cette fonction ne s'applique aux nombres mais pas aux matrices, on utilise la commande `map`.

```
> restart: with(LinearAlgebra):
```

```


A:=Matrix([[1,2],[2,-1]]);
P:=Matrix([[cos(a),-sin(a)],[sin(a),cos(a)]]);
Q:=Matrix([[cos(a),sin(a)],[sin(a),-cos(a)]]);
> B:=Transpose(P).A.P;
L:=solve(B[1,1]=0,a);
> f:= x->expand(rationalize(radnormal(x)));
a:=L[1];
map(f,B);

```

```

> a:=L[2]: map(f,B); a:=L[3]: map(f,B);
a:=L[4]: map(f,B);
> a:='a': #a redevient vierge
C:=Transpose(Q).A.Q;
L:=solve(C[1,1]=0,a);
> a:=L[1]: map(f,C); a:=L[2]: map(f,C);
a:=L[3]: map(f,C); a:=L[4]: map(f,C);

```

**Solution 7.**

Il est important de choisir un système de coordonnées adapté au problème.

```

> restart: # Repère orthonormé centré en M et d'axe des abscisses (AB)
# Equation du cercle et des droites
EqCercle:= x^2+(y-m)^2=1; EqPQ:= x=a*y; EqRS:= x=b*y;
> # Calcul des coordonnées de P ou Q en fonction de a :
L:=allvalues(solve({EqCercle,EqPQ},{x,y}));

```

On trouve les ordonnées de  $P$  et  $Q$ , puis on en déduit les abscisses.

```

> y[P]:=rhs(L[1][1]); x[P]:=rhs(L[1][2]); y[Q]:=rhs(L[2][1]); x[Q]:=rhs(L[2][2]);

```

Ensuite même chose pour  $R$  et  $S$ .

```

> # Calcul des coordonnées de R ou S en fonction de b :
K:=allvalues(solve({EqCercle,EqRS},{x,y}));
> y[R]:=rhs(K[1][1]); x[R]:=rhs(K[1][2]); y[S]:=rhs(K[2][1]); x[S]:=rhs(K[2][2]);

```

Finalement on calcule les abscisses de  $C$  et  $D$ .

```

> # Le point D est caractérisé par det(DQ,RQ)=0
x[D]:=solve((x[Q]-x)*(y[Q]-y[R])-(y[Q]-0)*(x[Q]-x[R])=0,x);
# Même chose pour C
x[C]:=solve((x[S]-x)*(y[S]-y[P])-(y[S]-0)*(x[S]-x[P])=0,x);

```

Enfin, on vérifie que l'abscisse de  $C$  et l'opposée de celle de  $D$ .

```

> simplify(x[C]+x[D]);

```

**Solution 8.**

1. On trouve  $[2, 1, 0, 1, 2]_3 = 194$  et

$$\begin{aligned}
 100 &= 1 \times 3^4 + 2 \times 3^2 + 1 \times 3^0 = [1, 0, 2, 0, 1]_3 \\
 &= 1 \times 4^3 + 2 \times 4^2 + 1 \times 4^1 = [1, 2, 1, 0]_4.
 \end{aligned}$$

2. On commence une boucle avec  $m = 0$  puis on l'augmente  $m$  tant qu'on a  $b^{m+1} \leq n$ .

```

> Maxi:=proc(n,b) local m:

```

```

m:=0: while n>=b^(m+1) do m:=m+1 od: m:
end proc:
> Maxi(26,3); Maxi(27,3); Maxi(2,3);
2, 3, 0

```

C'est pour assurer que le programme sort quelque chose aussi dans le cas  $n < b$  (boucle vide) qu'on a fait suivre la boucle par l'affichage de la valeur de  $m$ .

3.

```
> BaseEntier:=proc(n,b) local r,m,k,L:
  r:=n: m:=Maxi(n,b): L:=NULL:
  for k from 0 to m do
    L:=L,iquo(r,b^(m-k)): r:=irem(r,b^(m-k)):
  end do
end proc:
```

**Solution 9.**

Pour calculer le gradient et la matrice hessienne Maple propose la package `with(VectorCalculus)`, mais on peut faire les calculs aussi directement :

```
> restart:
g:=2*x^2-8*x*y^2-16/3*y^3+8*y^2-8;
gx:=diff(g,x); gy:=diff(g,y);
gxx:=diff(gx,x); gxy:=diff(gx,y);
gyy:=diff(gy,y);
h:=gxx*gyy-gxy^2;
```

On détermine les points stationnaires.

```
> s:=solve({gx=0,gy=0},{x,y});
```

Il y a trois points stationnaires :  $(0,0)$ ,  $(2,-1)$  et  $(\frac{1}{2}, \frac{1}{2})$ . On examine le signe des vecteurs propres de la matrice hessienne.

```
> assign(s[1]): gxx; h;
```

4, 64

Donc  $f$  est convexe au voisinage de  $(0,0)$  et admet un minimum local en  $(0,0)$ .

```
> x:='x': y:='y': assign(s[2]): h;
```

-192

Le point  $(2,-1)$  est un point selle (ou col) de  $f$ .

```
od: [L]:
end proc:
```

```
> BaseEntier(194,3); BaseEntier(194,10);
[2, 1, 0, 1, 2], [1, 9, 4]
```

```
> assign(s[3]): h;
```

-96

Le point  $(\frac{1}{2}, \frac{1}{2})$  est un autre point selle.

```
> x:='x': y:='y': with(plots):
contourplot(g,x=-.5..2.5,y=-1.5..1,contours=60);
```

